

Identification Division.

Program-ID. J7200551.

```
*
*
* * Copyright Wisconsin Department of Transportation
*
* * Permission is hereby granted, free of charge, to any person or
* * organisation to use this software and its associated files
* * subject to the following conditions:
*
* * The software may be redistributed free of charge to any other
* * person or organisation provided that the above copyright
* * notice, this permission notice and the disclaimer shall be
* * included with all copies of the Software.
*
* * THE SOFTWARE IS PROVIDED "AS IS", WITHOUT WARRANTY OF
* * ANY KIND, EXPRESS OR IMPLIED, INCLUDING BUT NOT LIMITED
* * TO THE WARRANTIES OF MERCHANTABILITY, FITNESS FOR A
* * PARTICULAR PURPOSE AND NONINFRINGEMENT. IN NO EVENT
* * SHALL THE AUTHORS OR COPYRIGHT HOLDERS BE LIABLE FOR
* * ANY CLAIM, DAMAGES OR OTHER LIABILITY, WHETHER IN AN
* * ACTION OF CONTRACT, TORT OR OTHERWISE, ARISING FROM,
* * OUT OF OR IN CONNECTION WITH THE SOFTWARE OR THE USE OR
* * OTHER DEALINGS IN THE SOFTWARE. IF THIS DISCLAIMER
* * CONTRADICTS LOCALLY APPLICABLE LAW THEN USE OF THIS
* * SOFTWARE IS PROHIBITED.
*
*
* * This program wraps the CEE3INF LE Callable Service such that
* * it is more friendly to high level programming languages.
*
* * Sample use:
*
* * Working-Storage Section.
* * 77 LE-INFO-PGM PIC X(008) VALUE 'J7200551'.
*
* * 01 WS-SYS-ENV.
* *   05 WS-SYS PIC X(032) VALUE SPACES.
* *   05 WS-ENV PIC X(032) VALUE SPACES.
* *   05 WS-LANG PIC X(032) VALUE SPACES.
*
* * 01 WS-LE-INFO.
* *   05 WS-LE-PRDT-NB PIC 9(005) VALUE ZEROES.
* *   05 WS-LE-VERS PIC 9(005) VALUE ZEROES.
* *   05 WS-LE-RLSE PIC 9(005) VALUE ZEROES.
* *   05 WS-LE-MDFN PIC 9(005) VALUE ZEROES.
*
```

```

* Procedure Division.
*   CALL LE-INFO-PGM USING
*     WS-SYS-ENV
*     WS-LE-INFO
*   END-CALL
*
*   IF RETURN-CODE = 0
*     IF WS-SYS(1:1) = '1'
*     OR WS-SYS(2:1) = '1'
*     OR WS-SYS(3:1) = '1'
*     OR WS-SYS(4:1) = '1' DISPLAY 'CICS' END-IF
*     IF WS-SYS(5:1) = '1' DISPLAY 'TSO ' END-IF
*     IF WS-SYS(6:1) = '1' DISPLAY 'BATCH' END-IF
*     IF WS-SYS(7:1) = '1' DISPLAY 'UNIX ' END-IF
*     EVALUATE WS-ENV(14:1) ALSO WS-ENV(15:1)
*       WHEN '0'          ALSO '0'          DISPLAY 'AMODE24'
*       WHEN '1'          ALSO '0'          DISPLAY 'AMODE31'
*       WHEN '1'          ALSO '1'          DISPLAY 'AMODE64?'
*     END-EVALUATE
*     DISPLAY 'LE Version ' WS-LE-VERS '.' WS-LE-RLSE
*       '.' WS-LE-MDFN
*   ELSE
*     error handling
*   END-IF
*
*
*

```

Environment Division.

Data Division.

Working-Storage Section.

```

01 CONSTANTS.
   05 MYNAME          PIC X(008) VALUE 'J7200551'.
   05 CEEMSG-DEST    PIC 9(008) COMP-5 VALUE 2.

```

Local-Storage Section.

```

01 WORK-AREAS.
   05 WS-RC          PIC 9(008) COMP-5 VALUE 0.
   05 BIT-SUB       PIC 9(008) COMP-5 VALUE 0.
   05 BIT-TO-TEST  PIC 9(008) COMP-5 VALUE 0.
   05 FULL-WORD-TO-TEST PIC 9(008) COMP-5 VALUE 0.
   05 WS-SYS        PIC 9(008) COMP-5 VALUE 0.
   05 WS-ENV        PIC 9(008) COMP-5 VALUE 0.
   05 WS-LANG       PIC 9(008) COMP-5 VALUE 0.
   05 WS-LANG-X
      REDEFINES
      WS-LANG.

```

```

    10 FILLER PIC X(003).
    10 WS-LANG-CEE3INF-VERS PIC X(001).
05 WS-LE-VERS.
    10 WS-LE-VERS-PP-X PIC X(001) VALUE LOW-VALUES.
    10 WS-LE-VERS-VV-X PIC X(001) VALUE LOW-VALUES.
    10 WS-LE-VERS-RR-X PIC X(001) VALUE LOW-VALUES.
    10 WS-LE-VERS-MM-X PIC X(001) VALUE LOW-VALUES.
05 WS-LE-VERS-PP-R.
    10 FILLER PIC X(001) VALUE LOW-VALUES.
    10 WS-LE-VERS-PP-LB PIC X(001) VALUE LOW-VALUES.
05 WS-LE-VERS-PP
REDEFINES
WS-LE-VERS-PP-R PIC 9(004) COMP-5.
05 WS-LE-VERS-VV-R.
    10 FILLER PIC X(001) VALUE LOW-VALUES.
    10 WS-LE-VERS-VV-LB PIC X(001) VALUE LOW-VALUES.
05 WS-LE-VERS-VV
REDEFINES
WS-LE-VERS-VV-R PIC 9(004) COMP-5.
05 WS-LE-VERS-RR-R.
    10 FILLER PIC X(001) VALUE LOW-VALUES.
    10 WS-LE-VERS-RR-LB PIC X(001) VALUE LOW-VALUES.
05 WS-LE-VERS-RR
REDEFINES
WS-LE-VERS-RR-R PIC 9(004) COMP-5.
05 WS-LE-VERS-MM-R.
    10 FILLER PIC X(001) VALUE LOW-VALUES.
    10 WS-LE-VERS-MM-LB PIC X(001) VALUE LOW-VALUES.
05 WS-LE-VERS-MM
REDEFINES
WS-LE-VERS-MM-R PIC 9(004) COMP-5.
05 WS-CEE3INF-VERS-R.
    10 FILLER PIC X(001) VALUE LOW-VALUES.
    10 WS-CEE3INF-VERS-LB PIC X(001) VALUE LOW-VALUES.
05 WS-CEE3INF-VERS
REDEFINES
WS-CEE3INF-VERS-R PIC 9(004) COMP-5.

01 SWITCHES.
05 BIT-TEST-RSLT-SW PIC 9(008) COMP-5.
    88 BIT-IS-ON VALUE 1.
    88 BIT-IS-OFF VALUE 0.
05 LE-FEEDBACK-CD PIC X(008) VALUE LOW-VALUES.
COPY CEEIGZCT.

01 LEFB-CD.

```

05	FST-DBL-WORD.	
10	FC-SEVERITY	PIC S9(004) COMP-5.
10	FC-MESSAGE	PIC S9(004) COMP-5.
10	FC-SEVERITY-CNTL	PIC X(001).
10	FC-FACILITY-ID	PIC X(003).
05	FC-I-S-INFO	PIC S9(009) COMP-5.
01	CEE3INF-LEFB-CD.	
05	FST-DBL-WORD.	
10	FC-SEVERITY	PIC S9(004) COMP-5.
10	FC-MESSAGE	PIC S9(004) COMP-5.
10	FC-SEVERITY-CNTL	PIC X(001).
10	FC-FACILITY-ID	PIC X(003).
05	FC-I-S-INFO	PIC S9(009) COMP-5.
01	CEE3DMP-LEFB-CD.	
05	FST-DBL-WORD.	
10	FC-SEVERITY	PIC S9(004) COMP-5.
10	FC-MESSAGE	PIC S9(004) COMP-5.
10	FC-SEVERITY-CNTL	PIC X(001).
10	FC-FACILITY-ID	PIC X(003).
05	FC-I-S-INFO	PIC S9(009) COMP-5.
01	CEEMSG-LEFB-CD.	
05	FST-DBL-WORD.	
10	FC-SEVERITY	PIC S9(004) COMP-5.
10	FC-MESSAGE	PIC S9(004) COMP-5.
10	FC-SEVERITY-CNTL	PIC X(001).
10	FC-FACILITY-ID	PIC X(003).
05	FC-I-S-INFO	PIC S9(009) COMP-5.
01	CEESITST-LEFB-CD.	
05	FST-DBL-WORD.	
10	FC-SEVERITY	PIC S9(004) COMP-5.
10	FC-MESSAGE	PIC S9(004) COMP-5.
10	FC-SEVERITY-CNTL	PIC X(001).
10	FC-FACILITY-ID	PIC X(003).
05	FC-I-S-INFO	PIC S9(009) COMP-5.
01	LCL-APLC-DEBUG-AREA.	
05	CEE3DMP-TITL.	
10		PIC X(010) VALUE '++++++='.
10	CEE3DMP-TITL-SPFC	PIC X(060) VALUE SPACES.
10		PIC X(010) VALUE '++++++='.
05	CEE3DMP-OPTIONS	PIC X(255) VALUE SPACES.

Linkage Section.

* Output - Execution "environment" consisting of switches
* indicating system (effectively the OS), subsystem
* (CICS, TSO, etc.), active language(s), whether
* not a Pre-Initialization Programming Interface
* (PIPI) environment is in effect and what type.

```
01 ENV-SUBSYS-FLAGS.  
  05 SYS-FLAGS.  
    10 SUBSYS-CICS-SW          PIC X(001).  
      88 SUBSYS-CICS          VALUE '1'.  
    10 SUBSYS-CICS-PIPI-SW    PIC X(001).  
      88 SUBSYS-CICS-PIPI     VALUE '1'.  
    10 SUBSYS-CICS-OTHR-1-SW  PIC X(001).  
      88 SUBSYS-CICS-OTHR-1   VALUE '1'.  
    10 SUBSYS-CICS-OTHR-2-SW  PIC X(001).  
      88 SUBSYS-CICS-OTHR-2   VALUE '1'.  
    10 SUBSYS-TSO-SW          PIC X(001).  
      88 SUBSYS-TSO           VALUE '1'.  
    10 SUBSYS-BTCH-SW         PIC X(001).  
      88 SUBSYS-BTCH          VALUE '1'.  
    10 SUBSYS-UNIX-SW         PIC X(001).  
      88 SUBSYS-UNIX          VALUE '1'.  
    10 FILLER                  PIC X(022).  
    10 SYS-ZVSE-SW            PIC X(001).  
      88 SYS-ZVSE             VALUE '1'.  
    10 SYS-ZOS-SW             PIC X(001).  
      88 SYS-ZOS              VALUE '1'.  
    10 SYS-ZOS-E-SW          PIC X(001).  
      88 SYS-ZOS-E            VALUE '1'.  
  05 ENV-FLAGS.  
    10 ENV-PIPI-SW            PIC X(001).  
      88 ENV-PIPI             VALUE '1'.  
    10 ENV-PIPI-MAIN-SW      PIC X(001).  
      88 ENV-PIPI-MAIN        VALUE '1'.  
    10 ENV-PIPI-SUB-SW       PIC X(001).  
      88 ENV-PIPI-SUB-SW     VALUE '1'.  
    10 ENV-PIPI-SUBDP-SW     PIC X(001).  
      88 ENV-PIPI-SUBDP-SW   VALUE '1'.  
    10 ENV-PICI-SW           PIC X(001).  
      88 ENV-PICI             VALUE '1'.  
    10 ENV-NESTED-ENCLAVE-SW PIC X(001).  
      88 ENV-NESTED-ENCLAVE  VALUE '1'.  
    10 ENV-LRR-ACTV-SW       PIC X(001).  
      88 ENV-LRR-ACTV        VALUE '1'.  
    10 ENV-RUNTIME-REUSE-SW  PIC X(001).  
      88 ENV-RUNTIME-REUSE   VALUE '1'.
```

```

10 ENV-XPLINK-SW          PIC X(001).
   88 ENV-XPLINK          VALUE '1'.
10 ENV-POSIX-SW          PIC X(001).
   88 ENV-POSIX          VALUE '1'.
10 ENV-PTHREAD-SW        PIC X(001).
   88 ENV-PTHREAD        VALUE '1'.
10 ENV-IPT-SW            PIC X(001).
   88 ENV-IPT            VALUE '1'.
10 ENV-MULTITHREADED-FORK-SW PIC X(001).
   88 ENV-MULTITHREADED-FORK VALUE '1'.
10 ENV-AMODE-SW-1        PIC X(001).
   88 ENV-AMODE-1        VALUE '1'.
10 ENV-AMODE-SW-2        PIC X(001).
   88 ENV-AMODE-2        VALUE '1'.
10 FILLER                 PIC X(017).
05 LANG-FLAGS.
   10 FILLER                 PIC X(003).
   10 LANG-C-CPP-SW         PIC X(001).
       88 LANG-C-CPP         VALUE '1'.
   10 FILLER                 PIC X(001).
   10 LANG-COBOL-SW         PIC X(001).
       88 LANG-COBOL         VALUE '1'.
   10 FILLER                 PIC X(001).
   10 LANG-FORTRAN-SW       PIC X(001).
       88 LANG-FORTRAN       VALUE '1'.
   10 FILLER                 PIC X(002).
   10 LANG-PLI-SW          PIC X(001).
       88 LANG-PLI           VALUE '1'.
   10 LANG-ENTP-PLI-SW     PIC X(001).
       88 LANG-ENTP-PLI     VALUE '1'.
   10 FILLER                 PIC X(012).
   10 CEE3INF-VERS         PIC 9(008).

01 ENV-TBL REDEFINES ENV-SUBSYS-FLAGS.
   05 SYS-BYTES             OCCURS 32 PIC X(001).
   05 ENV-BYTES             OCCURS 32 PIC X(001).
   05 LANG-BYTES           OCCURS 32 PIC X(001).

*   Output - information about LE itself.
01 LE-VERS.
*   LE Product number
   05 LE-PP                 PIC 9(005).
*   LE Version
   05 LE-VV                 PIC 9(005).
*   LE Release
   05 LE-RR                 PIC 9(005).

```

```
*      LE Modification level
05  LE-MM                PIC 9(005).
```

```
Procedure Division Using
ENV-SUBSYS-FLAGS
LE-VERS
```

```
.
```

```
*      Get information about our environment
CALL 'CEE3INF' USING
```

```
    WS-SYS
    WS-ENV
    WS-LANG
    WS-LE-VERS
    CEE3INF-LEFB-CD
```

```
END-CALL
```

```
MOVE FST-DBL-WORD OF CEE3INF-LEFB-CD TO LE-FEEDBACK-CD
IF CEE000
```

```
    PERFORM 0100-INIT
    PERFORM 1000-PRCS
```

```
ELSE
```

```
    MOVE CEE3INF-LEFB-CD TO LEFB-CD
    PERFORM 9998-LE-ERR
    MOVE 8 TO RETURN-CODE
```

```
END-IF
```

```
MOVE WS-RC TO RETURN-CODE
GOBACK.
```

```
0100-INIT.
```

```
INITIALIZE
    ENV-TBL
    LE-VERS
REPLACING
    ALPHABETIC BY SPACES
    ALPHANUMERIC BY ZEROES
    NUMERIC BY ZEROES
```

```
.
```

```
1000-PRCS.
```

```
PERFORM 1010-GET-SYS
```

```
IF WS-RC = 0
```

```
    PERFORM 1020-GET-ENV
```

```
END-IF
```

```
IF WS-RC = 0
    PERFORM 1030-GET-LANG
END-IF
```

```
PERFORM 1040-GET-VERS
```

```
.
```

```
1010-GET-SYS.
```

```
*
* Test each bit in the WS-SYS fullword to determine
* information about our operating environment.
*
* Some of these bits are currently (z/OS 1.10) reserved. We
* test them anyway because they may someday have meaningful
* data in them.
*
MOVE WS-SYS TO FULL-WORD-TO-TEST

PERFORM VARYING BIT-TO-TEST FROM 0 BY 1
UNTIL BIT-TO-TEST > 31 OR WS-RC NOT = 0
    PERFORM 8010-BIT-TEST
* Bits are numbered from the right, subscripts from the left
    COMPUTE BIT-SUB = 32 - BIT-TO-TEST
    EVALUATE TRUE
        WHEN BIT-IS-ON
            MOVE '1' TO SYS-BYTES(BIT-SUB)
        WHEN BIT-IS-OFF
            MOVE '0' TO SYS-BYTES(BIT-SUB)
    END-EVALUATE
END-PERFORM
.
```

```
1020-GET-ENV.
```

```
*
* Test each bit in the WS-ENV fullword to determine
* information about our operating environment.
*
* Some of these bits are currently (z/OS 1.10) reserved. We
* test them anyway because they may someday have meaningful
* data in them.
*
MOVE WS-ENV TO FULL-WORD-TO-TEST

PERFORM VARYING BIT-TO-TEST FROM 0 BY 1
UNTIL BIT-TO-TEST > 31 OR WS-RC NOT = 0
```



```

PERFORM 8010-BIT-TEST
* Bits are numbered from the right, subscripts from the left
COMPUTE BIT-SUB = 32 - BIT-TO-TEST
EVALUATE TRUE
  WHEN BIT-IS-ON
    MOVE '1' TO ENV-BYTES(BIT-SUB)
  WHEN BIT-IS-OFF
    MOVE '0' TO ENV-BYTES(BIT-SUB)
END-EVALUATE
END-PERFORM

```

1030-GET-LANG.

```

*
* Test the first 24 bits in the WS-LANG fullword to determine
* which languages are currently active. The last byte
* indicates which version of CEE3INF was called.
*
MOVE WS-LANG TO FULL-WORD-TO-TEST
MOVE WS-LANG-CEE3INF-VERS TO WS-CEE3INF-VERS-LB
MOVE WS-CEE3INF-VERS-LB TO CEE3INF-VERS

PERFORM VARYING BIT-TO-TEST FROM 8 BY 1
UNTIL BIT-TO-TEST > 31 OR WS-RC NOT = 0
  PERFORM 8010-BIT-TEST
* Bits are numbered from the right, subscripts from the left
  COMPUTE BIT-SUB = 32 - BIT-TO-TEST
  EVALUATE TRUE
    WHEN BIT-IS-ON
      MOVE '1' TO LANG-BYTES(BIT-SUB)
    WHEN BIT-IS-OFF
      MOVE '0' TO LANG-BYTES(BIT-SUB)
  END-EVALUATE
END-PERFORM

```

1040-GET-VERS.

```

MOVE WS-LE-VERS-PP-X TO WS-LE-VERS-PP-LB
MOVE WS-LE-VERS-PP TO LE-PP
MOVE WS-LE-VERS-VV-X TO WS-LE-VERS-VV-LB
MOVE WS-LE-VERS-VV TO LE-VV
MOVE WS-LE-VERS-RR-X TO WS-LE-VERS-RR-LB
MOVE WS-LE-VERS-RR TO LE-RR
MOVE WS-LE-VERS-MM-X TO WS-LE-VERS-MM-LB
MOVE WS-LE-VERS-MM TO LE-MM

```

8010-BIT-TEST.

```
* Call LE service to test bits
CALL 'CEESITST' USING
    FULL-WORD-TO-TEST
    BIT-TO-TEST
    CEESITST-LEFB-CD
    BIT-TEST-RSLT-SW
END-CALL
```

```
MOVE FST-DBL-WORD OF CEESITST-LEFB-CD TO LE-FEEDBACK-CD
IF CEE000
    CONTINUE
ELSE
    MOVE CEESITST-LEFB-CD TO LEFB-CD
    PERFORM 9998-LE-ERR
    MOVE 8 TO WS-RC
    STRING MYNAME
        ' unrecognized return value from CEESITST'
        DELIMITED SIZE
        INTO CEE3DMP-TITL-SPFC
    END-STRING
    PERFORM 9997-DUMP-CORE
END-IF
```

.

9997-DUMP-CORE.

```
* Create a core dump to assist in debugging problems
CALL 'CEE3DMP' USING
    CEE3DMP-TITL
    CEE3DMP-OPTIONS
    CEE3DMP-LEFB-CD
END-CALL
```

.

9998-LE-ERR.

```
* Display LE message associated with feedback code
CALL 'CEEMSG' USING
    LEFB-CD
    CEEMSG-DEST
    CEEMSG-LEFB-CD
END-CALL
```

.